
PSL Programming
Introduction
PSL is first and foremost, a programming language.  The saving grace is that it is a simple and intuitive one.  If you have ever heard anyone claim that programming a computer is only for people with thick glasses and monumental grade point averages, don’t believe it!  If you can type, you can write a PSL program.
This document contains numerous short program fragments as examples.  They appear in bold face courier font.  As you read the document, feel free to copy/paste the examples into the PSL editor and try them out.

Hello World

The standard example given in every introduction to programming, whatever the language, always seems to be a program that prints “Hello world!” to some output or other.  In the case of PSL that would mean we want to write a program to generate a PDF document containing that phrase.  Here is the program:

Hello world!
Type this into the PSL editor and press the airplane button.  Not exactly tough, was it?  The simplest PSL program is just plain old text.  Whenever PSL sees some text, it sends it to the PDF output.  Essentially, any text file you happen to have is already a PSL program that generates a PDF version of itself.

Commands

Let’s examine the PDF output of the “Hello world!” program a little more closely.  The phrase appears at the top left corner of a “letter” (8 ½ by 11 inch) sized document presented in “portrait” mode.  The text is black and it is printed in a 12 point “times-roman” font.  These details are all PSL “default” values for things that can be changed.  A “default” value is just an arbitrary (hopefully reasonable) choice that PSL makes for something that the programmer has not explicitly specified.  All that we need do to change any of these things, is to add the appropriate PSL command.  PSL commands are just text, for instance the color of the PDF output can be changed using the “setColor” command:

setColor red

Wait a minute, isn’t there a problem here?  If you type this into the PSL editor we already know what will happen.  It will just generate a PDF file that says “setColor red” at the top left!  We need a way to tell PSL that this particular bit of text is a command rather than run of the mill text.  This is done by placing the text for the command in brace brackets.  When PSL sees ‘{‘, it takes everything following it as command text, up to the closing ‘}’.  Try this:
Hello {setColor red}world!

The output looks the same as before except that the color of the “ink” has changed to red between the words “Hello “ and “world!”.  Any amount of PSL command text can be embedded anywhere in the “regular” text by enclosing it in braces.  There is no limit to how many commands may be so enclosed, PSL will execute the commands as it encounters them.  We refer to the parts of a PSL program enclosed in braces as “code” sections.  (“code” is just the “techie” word for program text)  Not surprisingly, the parts outside the braces are known as ”text” sections.
So how do I print a brace bracket?  Interesting problem, there is a standard answer.  We use what is known as an “escape” character.  Whenever PSL sees a character in the text preceded by a backslash (’\’) it ignores the backslash and uses the following character “as is”.  So if we want a brace, we just stick a backslash in front of it.  Try this:
Hello \{setColor red\}world!  And how about a backslash? \\

We always ignore backslashes in text, as \we \can \s\e\e here.

Let’s try adding a few other commands to doll up the “Hello world!” example a bit:
{

  setPaperOrientation landscape

  setFont "Helvetica-bold", 50

  setColor blue

}

Hello world!
We have used three commands to change some of those “default” values that PSL uses.  You can read more about the effects of any of these commands by looking them up in the PSL reference manual.  As long as the commands are enclosed in braces, PSL does not care about what format they are typed in.  It simply executes them in the order in which they are typed.  We could just as well have used:
{setPaperOrientation landscape setFont "Helvetica-bold",50 setColor blue}Hello world!
When reading commands, PSL ignores extra “white space” characters like tabs, spaces and new lines.  We can take advantage of this to format a series of commands to make them easier to read.  In fact, all the examples this document are formatted using a simple convention based on this idea.  If we are using just one command, we often embed it directly in the text:
Hello {setColor red}world!

On the other hand, if using more than one in a series, we prefer to place them on separate lines and indent them to make them stand out from the surrounding text:

{

  setPaperSize 10, 20
  setFont "times-bold", 100
  setColor blue
  setBackground yellow
}

What a long

piece of paper!

It is always worth while taking advantage of conventions like this to make programs easier to read. For instance, PSL does not care about character case in commands, we add the upper case characters for no other reason than that “setPaperSize” seems a lot easier to read than “setpapersize”.  Get into the habit of using conventions like these, the extra typing effort pays off “big time” when it comes to dealing with something you or someone else wrote last month!
Statements

To be a bit more precise, the things that can appear between brace brackets are not just commands, but PSL “statements”.  There are several types of statement that PSL understands, the “commands” we have been using are examples of one such type.  We will examine each type in turn.  Before doing that, there is one rather useful command that is worth knowing about, since we will be using frequently in our examples.  This is the “show” command.  The “show” command has one argument, which can be any value.  The effect of the command is to display the value on the screen.  If you are using the PSL editor, it will be displayed in the status window at the bottom of the editor, if you are using the command line version, it will appear in the DOS box.  Try this example:
{

  show "Hello world!"
}
When you run this program, the phrase will appear (minus the quotes) in the status window instead of in a PDF document.  We will make extensive use of this command to illustrate the actions of various statement types by sending output to the status window.

Assignment Statements and Data Types
Like any programming language, PSL deals with data of one sort or another.  In order to do so, we need a way of naming data items so that we can tell PSL which one we are referring to at any particular time.  We can associate a name with a value using an “assignment” statement.  The assignment statement is just a name followed by “=” and the desired value:

{

  someName = 1

}

This statement assigned the value 1 to a “variable” called “someName”.  Try running the following example:

{

  age = 21

  show age

}

The named variable “age” was assigned the value 21.  Thereafter, the name “age” is a valid value which may be used as an argument to commands like “show”.  We can change the value by means of another assignment statement:
{

  age = 21

  show age

  age = 99

  show " changed to "
  show age

}
We have already seen examples of two different types of data that PSL can deal with.  These are “Numbers”, such as 21, and “Strings” of characters, such as “Hello World!”. 
 Numbers may have a decimal point and a sign if desired.  Some examples of valid Numbers are:

21, 3.14159, -1, +42, .5
In order to permit very large (and very small) values, Numbers may also have an “exponent”, indicating that the value is to be multiplied by a power of ten:


1.1E2

1.1 times ten squared = 1.1 times 100 = 110


1.1E-3

1.1 times ten to the minus 3 = 1.1 divided by 1000 = 0.0011


1.1E120
a very large number indeed


1.1E-120
and a very, very small one

Data of type “String” is just text, a “string of characters”.  You tell PSL where it begins and ends with the double quote character:

"Hello World!”,  “21”,  “My name is Charley”

These are all valid examples of String data.  Note that “21” is NOT the same as 21, the first is a String, the second a Number.

What if we want to include a double quote character in a String?  Remember the escape character (backslash) we used to allow special characters in text?  Just use that:
{
  show "He said \"Hello there\", and sat down."
}

In total, PSL recognizes eleven different types of data.  You can read about them in the reference manual in the “data Types” section.  Here is one that we have seen before:

{setColor red}Hello World!

This is an example of data type “Color”.  Since PSL is concerned with typesetting it should not come as a great surprise that it treats colors as a type of data!  To learn more about this data type, you can read the chapter on color control. (colorControl.doc)  Suffice it to say that it is perfectly reasonable to say:
{

  c = red

  show c

}

You will see that displayed result of the “show” command indicates that “red” is, in fact, an “RGB” color with “component” values (red, green, blue) of (1,0,0).

What is the value of a “variable” before we assign it a value?  It would clearly be a mistake to use one without having first assigned it a value!  Indeed, this is such a common error that PSL has a special data type designed to detect it.  As you might guess, there are virtually an infinite number of variables (as many as there are names for them).  PSL effectively pre-assigns them all with a special value type known as “Uninitialized”.  It will report an error if your program attempts to make use of such a value:
{

  age = anUninitializedVariable

}

The error message in the status window gives the line number in the program where the error occurred, in this case “L 2”, indicating line number 2.  The show command will display such a value as the word “uninitialized”:
{

  show anUninitializedVariable

}

Expressions
One would expect that a language meant for programming a computer should be able to do simple arithmetic!  How do we tell PSL to do it?  The answer is, pretty much the same way we write it on paper:

{

  x = 1

  y = 2

  show x + y

}

The “x + y” is an example of what is known as an “expression”.  Expressions use the pretty much same rules as your teacher taught you for arithmetic.  They are made up of values, such as 21 or “x”, and “operators”, such as + or -.  They can also use parenthesis to change the order in which the operations are done.  Normally, arithmetic operations are done in the same order the teacher said, that is, multiply and divide first, then add and subtract.  The parenthesis may be used to override this “operator precedence” scheme:
{

  show 2 * 3 + 1

  show " is not the same as "
  show 2 * (3 + 1)

}

PSL has a somewhat larger set of operators than just the ones for arithmetic, and expressions can deal with data types other than numbers.  The arithmetic operators are (in order of highest to lowest “precedence”):


^
power, eg 2^2 is 2 to the power 2 = 4

*,/
multiplication and division


+,-
addition and subtraction

We can also “add” two Strings together using the “concatenation” operator ‘&’.  This has the effect of adding the second String on to the end of the first:

{

  show "Hello " & "there!"
}

PSL also has a set of “comparison” operators:


=
equal


<>
not equal


<
less than


> 
greater than


>=
greater than or equal


<=
less than or equal

What is the result of applying such an operator to two values?  The answer is, of course, “true” or “false”.  PSL does not have a special data type for these values.  Instead, it uses Numbers.  The value for “false” is zero, any other Number is interpreted as representing “true”.  These “Boolean” operators all produce the value one for “true” and zero for “false”.  As we will see below, there are a number of PSL statements that make use of this type of value.  In fact, they are so useful that PSL recognizes the words “true” and “false”:
{

  x = 1

  y = 2

  show x = y

  show " "
  show x < y

  show " "
  show true

  show " "
  show false

  show " A comes before B "
  show "Aardvark" < "Bear"
}

In the case of Strings, the comparison operators perform an alphabetic comparison.  That is, “A” is less than “B” because it comes before “B” alphabetically.

There are also three operators that work directly with these “boolean” or “logical” values:


not
changes true to false and false to true


and
true only if both values are true


or
true if either or both values are true

{

  show not true

  show newline

  show true and false

  show " "
  show true and true

  show newline

  show true or false

}

Note that we have made use of a handy value called “newline”.  This is a String containing a special character that causes the output to move to the next line on the display.

For more detailed information on the behavior of operators and values in expressions, see the PSL reference manual under “Grammar/Statements/Expressions”.

When we referred earlier to a “value”, what we really meant was an “expression”!  In a PSL statement, an entire expression, however long and complicated, may be placed wherever a value is needed.
Finally, consider the somewhat unusual looking assignment statement in this example:

{

  x = 1

  x = x + 1

  show x

}

The first statement assigns the value 1 to “x”.  The second assigns it to the value of the expression “x + 1”.  Since that is 2 at the time the expression is “evaluated”, the resulting value of “x” is 2.  There is actually nothing unusual in this, the statement “x = x + 1” is NOT an equation, it is just an instruction to PSL to assign a value to the name “x”.
If Statement
The title above is not a question, it is the name of a statement!  The “If” statement is a statement type that allows a PSL program to make decisions.  Essentially, it executes a statement only if a value is true.  Otherwise it ignores the statement.  The form of an “If” statement is:


“if” expression statement

For example:

{

  x = 1

  if x < 2

    show "Yes it is smaller than 2"
  if x > 2

    show "Hmm, it is larger than 2"
}

The first “if” statement executes the statement that displays "Yes it is smaller than 2" providing the value of “x” is less than 2.  The second will display "Hmm, it is larger than 2" if “x” is greater than 2.  Try it again with a different value of “x”:

{

  x = 3

  if x < 2

    show "Yes it is smaller than 2"
  if x > 2

    show "Hmm, it is larger than 2"
}

Notice that we have placed the statement that is the object of the “if” on the next line and indented it a bit.  Recall that PSL does not care about extra new lines or spaces, we could just as well have written:
{

  x = 3

  if x < 2 show "Yes it is smaller than 2"
  if x > 2 show "Hmm, it is larger than 2"
}

or even:

{x = 3 if x < 2 show "Yes it is smaller than 2" if x > 2 show "Hmm, it is larger than 2"}
PSL doesn’t care.  However, the indented version is easier to read, we have used the indentation to emphasize the fact that the statement with the “show” command is subject to the “if”.  This may seem trivial at this point but it becomes pretty valuable when programs get a little more complex.

What happens if we want to execute more than one statement when a condition is true?  Do we have to write something as cumbersome as:

{

  x = 2

  if x < 3

    show "x is actually "
  if x < 3

    show x

}

This could get a little annoying!  Instead, we use a construct called a “compound statement”.  We can collect up any number of statements into a single effective statement just by beginning and ending them with the words “do” and “end”:

{

  x = 2

  if x < 3 do

    show "x is actually "
    show x

  end

}

Again, PSL does not care how we format the placement of the “do” and “end”, we have just chosen a convention that makes the program easier to understand.
Since the subject of an “If” statement is another statement, there is nothing wrong with making that statement another “If”:

{

  x = 2

  y = 3

  if x < y

    if y = 3

      show "x is less than 3"
}

“If” statements may be “nested” in this way to any depth required.  In general, this is true of any statement type that includes another statement.  Such “recursive” behavior is one of the things makes programming languages based on a few simple statement types as powerful as they are.  One can build statements as deep and complex as required from just a few simple constructs.

Our definition of the “If” statement given at the start of this section was, in fact, incomplete.  The full definition is actually:

“if” expression statement (“else” statement)

The part of the definition inside the parenthesis is optional.  If the optional “else” part is used, the statement associated with the “else” will be executed whenever the condition is NOT true:
{
  x = 2

  if x < 1

    show "x is less than 1"
  else

    show "x is not, in fact, less than 1"
}
There is, of course, nothing to stop us from stringing these together in any way we like:

{

  x = 2

  if x < 1

    show "x is less than 1"
  else if x < 2

    show "x is less than 2"
  else

    show "x is neither less than 1 nor less than 2"
}

Whenever PSL “sees” an “else”, it associates it with the most recent “if”.

Loop Statements
A common programming requirement is to execute a statement repeatedly, stopping only when some condition is met.  PSL has three statement types that will do this.  The most basic of these is the “while” statement:


“while” expression statement

This will execute the “statement” as long as the “expression” has the value “true” (that is, is non-zero).  Clearly the statement that is the body of the loop had better change something that affects the value of the expression!  The body of a “while” statement is usually a compound statement.  This program will display the numbers 1 through 10:

{

  x = 1
  while x < 11 do

    show x & " "
    x = x + 1

  end

}

We have used the String concatenation operator ‘&’ to combine the value of “x” and a space (“ “) into one expression for the “show” command. 

Note that if the expression has the value false when the “while” statement is first encountered, the body of the loop will not be executed at all.  In short, the condition is evaluated BEFORE the body is executed.  If we wish to make a loop where the body always executes at least once, we would need to evaluate the condition AFTER the body is executed.  This version of a loop is called a “repeat” statement:


“repeat” anyNumberOfStatements “until” expression

The statements will be repeatedly executed as long as the expression evaluates to false.  Printing numbers up to ten now looks like:
{

  x = 1
  repeat

    show x & " "
    x = x + 1

  until x > 10

}

The body of a “repeat” statement will always execute at least once even if the condition starts out as true:

{

  x = 99
  repeat

    show x & " "
    x = x + 1

  until x > 10

}

Finally, it is often the case that we wish to execute a statement over some regular range of numeric values.  This requirement is so common that many languages, including PSL, provide a special form of loop statement just for that purpose:



“for” name “=” expression1 “to” expression2 (“step” expression3) statement

This “for” statement starts by initializing the value of “name” to that of “expression1.  If that value is less than or equal to the value of “expression2” it executes the “statement”.  Then it adds “expression3” to the value of “name” and tests again.  If the optional “step” clause is not provided it adds the value 1.  The statement will continue executing as long as the value of “name” does not exceed the value of “expression2”.  The “for” statement is essentially just shorthand for a common form of “while” loop:

name = expression1


while name <= expression2 do


    statement


    name = name + expression3


end

For instance, our program to print 1 to 10 is now just:

{

  for x = 1 to 10

    show x & " "
}

The program is much more compact and easier to read with the “for” statement than with the “while”.  If we use the “step” option we can easily change to say, odd numbers only:
{

  for x = 1 to 10 step 2

    show x & " "
}

Since statements can be nested it is pretty easy to generate a lot of stuff with “for” loops:
{

  for i = 1 to 10 do

    for j = 1 to 10

      show j & " "
    show newline

  end

}

Exit and Continue Statements

What happens if we need to exit from a loop before the all the statements in the body have executed?  PSL provides two statements expressly for the purpose of modifying loop execution.  The “exit” statement stops execution of a loop:
{

  x = 1

  y = 7

  while x < 11 do

    show x & " "
    x = x + 1

    if x > y

      exit

  end

}

In this case we have terminated the loop when the value of “x” exceeded that of “y”, even though the loop condition was still true. Of course, “exit” statements may only appear within the body of a loop, PSL will report a syntax error otherwise:

{

  x = 1

  exit  % oops, an error

}

Something new has been added here.  PSL takes everything from a ‘%’ to the end of the line (or code section) as a “comment” and ignores it.  We use such “comments” to add commentary to the program as an aid to understanding.

PSL also provides a sort of “opposite” to the “exit” statement.  The “continue” statement causes execution to “jump” back to the top of the loop.  In the case of a “for” loop, the counter value will be incremented first:

{

  y = 7

  for x = 1 to 10 do

    if x = 7

      continue

    show x & " "
  end
}

In this case we skip the remaining statements when “continue” is executed, thus omitting the 7 in the output.

WARNING: Always be careful when using “continue” statements, it is all too easy to create a loop that never exits!  The computer is, after all, only a machine, it will behave true to form and “brainlessly” keep executing such a loop until the end if time!  For example, adding a “continue” to our “while” example:

{

  % DANGER! DO NOT RUN THIS ENDLESS PROGRAM!

  x = 1

  y = 7

  while x < 11 do

    if x = y

      continue

    show x & " "

    x = x + 1

  end
}
Here is an interesting problem.  Recalling that we can nest loops, how do we exit from the outer one when we are in the inner?  The “exit” and “continue” statements both normally apply to the “closest” loop.  For instance:
{

  y = 7

  for i = 1 to 10 do

    show newline

    for j = 1 to 10 do

      % How do I exit the whole thing from here?

      show j & " "
    end

  end

}

An “exit” at the point where we asked the question would exit from the inner (“j”) loop, not the outer one counting on “i”.   PSL allows us to give a name to a loop and provides “named” forms of the “exit” and “continue” statements.  Any loop may be preceded by a name followed by a colon, which names the loop.  The full definition of the exit and continue statements, is actually:

exitStatement = “exit” (“from” loopName)


continueStatement = “continue” (“with” loopName)

So our example with the nested “for” loops can be written as:

{

  y = 7

  outer : for i = 1 to 10 do

    show newline  
    for j = 1 to 10 do
      if j = y

        exit from outer

      show j & " "
    end

  end

}

Similarly for the “continue” statement:

{

  y = 3

  outer : for i = 1 to 10 do

    show newline

    for j = 1 to 10 do

      if j = y

        continue with outer

      show j & " "
    end

  end

}

Stop Statement
This is the simplest one going!  The “stop” statement does just what it says, it stops execution of the program.  For example:

{

  show "Hello there" & newline

  stop

  show "Never get here"
}

As expected, the message “Never get here” is not displayed since the program was stopped before the statement that displays it was reached.

NOTE: PSL programs are often run with “merge” data.  In that case the program executes once for each record in the merge file.  A “stop” statement stops the current execution, NOT the entire system.  It will stop the execution for the current merge record and the program will start over on the next one.

Select Statement
It often happens that we wish to choose a statement to execute based on some value.  For example, we might want to take a different action depending on the city in which a customer lives.  It is, of course, possible to write such a program using a chain of “if” statements, but that is somewhat cumbersome and definitely hard to read.  Instead, PSL provides a “select” statement, which “selects” a statement to execute based on a value.  The definition of the select statement in the reference manual is a bit complex looking, easier to give an example or three:

{

  city = "New York"
  with city select

    "New York" :

      show "The Big Apple"
    "San Francisco" :

      show "Frisco"
    "Los Angeles" :

      show "LA"
    "Toronto" :

      show "TO"
    otherwise :

      show "Nowheresville"
  end

}

When this example runs, it uses the value of “city” to select a statement to execute.  The “select” statement contains a list of options, each consisting of a value followed by a colon and a statement.  There is also an optional “otherwise” option.  When the “select” statement is executed it simply goes down the list looking for a match to the value of “city”.  If it finds one, it executes the corresponding statement.  If it does not find a match, it will choose the “otherwise”, if there is one, and do nothing if there is not.  If more than one value matches it will choose the first one.
In general, the selection options may contain more than one value, if so, the values must be separated by commas:

{

  city = "Los Angeles"
  with city select

    "New York" :

      show "The Big Apple"
    "San Francisco", "Los Angeles" :

      show "Somewhere in California"
    "Toronto" :

      show "TO"
    otherwise :

      show "Nowheresville"
  end

}

Finally, both the selection value and the option values may be arbitrary expressions if desired:
{

  x = 2

  y = 2

  z = 3

  with x + y select

    1,2 :

      show "sum is one or two"
    z :

      show "sum is the same as z"
    z + 1 :

      show "sum exceeds z by one"
  end

}
As usual, there is nothing wrong with using compound statements and/or having select statements in select statements, however deep.
Function Calls
A function is just a named sub-program that “returns” a value.  For example, if we want to display the trigonometric sine of 45 degrees we use the built-in “sin” function:

{

  show sin 45

}
The function has a name, “sin”, and, in this case, an argument, the angle.  It also has a value, the sine of the argument, and therefore can be used in an expression, like any other value.  When PSL sees a function name in an expression, it runs the appropriate sub-program and places the value it generates into the expression in place of the function name and argument values.  In this case it replaced “sin 45” in the expression with the value generated by the “sin” sub-program for argument value 45.

In general, all functions have a name and require some number of arguments, the number depending on the specific function.  They all have a value, again, depending on the specific function.  In addition, they can also perform other work, such as generating PDF output or changing PSL “default” values.
PSL has a large number of such functions built in to it.  In fact, most of the reference manual is devoted to listing these functions and the actions they perform.  We can run a specific function by means of the “function call” statement:

functionCall = functionName (argumentList)

where:


argumentList = expression (“,” argumentList)
The second definition is rather interesting.  The definition of “argumentList” contains a reference to itself!  This is nothing more than a compact way of saying that the list can have any number of argument values, separated by commas.  Some examples of valid function call statements are:


newline


sin x + y


setPaperSize 8.5, 11

When we referred to “commands” in our earlier discussion, what we were actually talking about were just function call statements.  Even the “newline” command is a function call!  It runs a subprogram that “returns” a String containing the new line character.
Function Definitions
As mentioned above, PSL has a large number of useful built-in functions.  However, that is not the end of the story for functions.  There is, of course, no way that any system can provide every function that might be useful, the number would be infinite!  Instead, PSL provides a means to define new functions as required.  This means that you can create your own “custom” commands as required for any particular purpose.  Such “function definitions” are the means by which a programming language such as PSL may be extended for any particular application.  They also provide a handy way to split up large programs into smaller, more easily understood, sub-programs.  Often, writing a program amounts to a process of defining and writing an appropriate set of such sub-programs.  A function definition is:

“function” functionName (argumentNameList) “do” (statementList) “end”
where:


argumentNameList = name (“,” argumentNameList)

and:


statementList = statement (statementList)

A few examples are in order.  Here is a simple function with no arguments, together with a function call statement to make use of it:

{

  showHelloThere
% function call

  % The function definition, just does what the name says

  function showHelloThere do

    show "Hello there" & newline
  end

}

It generally pays to give functions names that relate to what they do, if at all possible.  This one just displays the String “Hello there”.  We usually precede a function definition with a comment that explains the purpose of the function.  
We mentioned above that all functions have a “return” value.  That is, if “called” in an expression, they generate a value that replaces the function call.  To define the “return” value in a function definition we simply include an assignment statement where the name to which the value is assigned is the name of the function:
{

  show onePlus 5

  % onePlus x

  %   Returns "x" plus 1

  function onePlus x do

    onePlus = x + 1

  end

}

We can have more than one such return value assignment, the value that is actually returned will be the last one executed:

{

  show roundOff 5.2
  % roundOff x

  %   Returns the nearest integer value to "x"
  %   Only works properly for positive numbers

  function roundOff x do

    integerX = integerPartOf x  % integerPartOf is a built-in function
    roundOff = integerX

    if x - integerX >= 0.5

      roundOff = integerX + 1

  end
}

NOTE: Although we can assign a value to the function name within the definition, the name remains a FUNCTION name, it is NOT a variable.  An attempt to use the value by that name inside the definition will be interpreted as a function call.

So, if functions always have a value, what is the return value of our first example?  It contained no such assignment.  Can we use it as a value?  In PSL, ALL functions have a return value.  If the definition does not assign one, PSL provides a default.  Try using our first function as a value:
{

  show showHelloThere
% show the default value

  % The function definition, just does what the name says

  function showHelloThere do

    show "Hello there" & newline

  end

}

The value “nil” is a special data type that indicates “no data”.  Note that this is NOT the same as “Uninitialized”, there is nothing wrong with assigning the value “nil” to a variable.  All PSL functions return “nil” by default, they act as though the first line of the function were:

functionName = nil

Recalling that statements can contain instances of themselves to any depth, can we do the same with function definitions?  The answer is, “sort of, in an interesting way”.  Function definitions in PSL can not contain other function definitions, but they can contain statements that call the function being defined.  Such function definitions are called “recursive”, the classic example is a function that calculates the “factorial” of a number:
{

  show factorial 5

  function factorial n do

    assert n >= 0, "Invalid argument for function factorial"
    if n = 0

      factorial = 1

    else

      factorial = n * factorial n - 1

  end

}

Have a look in the manual under “Built In Functions/Console Output Functions” to see what “assert” does.

When the factorial function calls itself in this example, the argument value is different.  Indeed, it is easy to see that this definition is liable to wind up calling itself with quite a number of different values.  What happens if we try to use the argument AFTER the recursive call?  Try this slightly modified example:

{

  show factorial 5

  function factorial n do

    assert n >= 0, "Invalid argument for function factorial"
    if n = 0

      factorial = 1

    else

      factorial = n * factorial n - 1

    show "factorial " & n & newline

  end

}

This example allows us to “trace” the progress of the calculation.  It also illustrates the fact that each invocation of “factorial” has its own independent copy of the argument value.  Function arguments in PSL are “passed” by value.  That is, when we provide a value for an argument, PSL makes a new copy of it and passes the copy to the function, rather than the original value.  This has the effect of preventing functions causing trouble by changing values in the rest of the program, to say nothing of keeping things straight in recursive functions like “factorial”.
WARNING: As you might expect, recursive functions are great candidates for making “endless loop” programs.  Be careful when using them!
Return Statement
Recall that PSL provides “exit” and “continue” statements to avoid execution of portions of the statements in a loop body.  The “return” statement serves the same purpose in a function body.  It simply causes the function to stop and “return” to point in the program that called the function.

{

  countToLeastOfTenOr 5

  % countToLeastOfTenOr x

  %   Shows the integers from 1 to 10 or x,

  %   whichever is smaller

  function countToLeastOfTenOr x do

    for i = 1 to 10 do

      if i > x

        return

      show i & newline

    end

    show "I got all the way to ten!"
  end

}

The program will not print “I got all the way to ten!” unless the argument is 10 or greater, as it would have, had we used “exit” instead of “return”.

